FIALA Tibor

A DÍJTARTALÉK ÉS A BIZTOSÍTÁSI DÍJ KOMPONENSEI

A tanulmány egy számítástechnikailag jól kezelhető modellt ismertet egyrészt a díjtartalék, másrészt a biztosítási díj két komponense (kárenyhítésre fordítandó rész és tartalékolt rész) közötti öszefüggés leírására, valamint – ebből következően – a függő változóként szabadon választott elem számszerű meghatározására.

Életbiztosítási szerződéseknél tipikus, hogy a futamidő elején a díjfizetés intenzív, a biztosítónál felhalmozódik a pénz. A futamidő vége felé díjbefizetés már általában nincs, viszont egyre nagyobb valószínűséggel lesz szükség a biztosítási összeg kifizetésére. Ezért a befizetett díjaknak csak egy része járul hozzá az adott évben például haláleset miatt bekövetkező kifizetések fedezetéhez, a díjak másik részét a biztosító későbbi kifizetéseinek teljesítése céljából tartalékolja. A tartalékolt részekből gyűlik össze az ún. díjtartalék, aminek a vizsgálata jelen munkánk egyik fontos célkitűzése. A díjtartalék szoros összefüggésben van a a befizetett díjak két komponensével, tehát az adott évben kárenyhítésre fordítandó résszel és a tartalékolt résszel. Ezeknek az összefüggéseknek a részletes elemzése képezi dolgozatunk fő tárgyát.

Vizsgálatainkat az ún. *vegyes életbiztosítás* esetére fogjuk elvégezni, ami a következőképpen szól.

x éves férfi köti a biztosítást. Ha a férfi n éven belül meghal, akkor a biztosító fizet a kedvezményezettnek egységnyi pénzt (pl. 1 millió Ft-ot). Ha viszont a biztosított n év múlva életben van, akkor ő kap a biztosítótól S összeget (pl. 400.000 Ft-ot). Ennek fejében a biztosított biztosítási díjat fizet a biztosítónak, minden év elején azonos összegben, k éven keresztül. Kérdés, hogy mekkora legyen ez a díj, hogyan alakulnak ennek a komponensei, és hogyan alakul a biztosítónál felhalmozott díjtartalék?

Mindenekelőtt pontosítani kell a feladatot. A kezdeti időpont a biztosítás megkötésének időpontja. Ekkor a biztosított befizeti az első díjat. A következő díjat pontosan egy év múlva fizeti, aztán pontosan 2 év múlva, végül az utolsót pontosan k-1 év múlva, ha még életben van. Ha korábban meghal, senki nem fizeti be helyette a hátralévő díjakat. Balról zárt és jobbról nyitott évekkel dolgozunk, tehát azt mondjuk, hogy a biztosított az i-edik évben halt meg, ha a halál a kezdeti időpont után legalább i évvel, de még az i+1-edik év kezdete előtt következik be. Ha a halál a nulladik évben következik be, akkor a biztosító az első év elején fizet, ha a halál az i-edik évben következik be, és i<=n-1, akkor a biztosító az (i+1). év elején fizet. Végül abban az esetben, ha a biztosított az n-edik év elején életben van, akkor a biztosító az n-edik év elején kifizeti az S összeget, és ezután már további kötelezettsége nincsen. Egyszerűen fogalmazva: év elején fizet a biztosított, s ha év közben meghal, akkor a következő év elején fizet a biztosító. Természetesen k<=n, hiszen nem várható el a biztosítottól, hogy olyankor is fizessen, amikor már biztosan nem kap semmit cserébe.

Az egyszerűség kedvéért eltekintünk a biztosító költségeitől, tehát úgy gondolkodunk, hogy a biztosító ingyen szedi be a díjakat, vezeti a nyilvántartást, és külön költség nélkül folyósítja a szükséges biztosítási összegeket. Ez azt jelenti, hogy az ún. *nettó díjat* számoljuk ki, ennek a komponenseit és a *nettó díjtartalékot* vizsgáljuk.

A biztosítási díj meghatározása

A szükséges számításokat a Magyarországon legelterjedtebb táblázatkezelő programmal, az EXCEL-lel fogjuk elvégezni. A Microsoft Excel 5-ös verziója, vagy bármelyik újabb verzió használható. Szükségünk lesz egy ún. életbenmaradási statisztikára, ami azt mutatja, hogy 100000 újszülöttből hányan érik el az 1 éves, 2 éves, ... , 100 éves kort. A következő táblázat például az 1988-as férfi életbenmaradási statisztikát mutatja Krekó Béla [5] műve szerint. (Az adatok lemezen is megtalálhatóak a szerző könyvének [2] lemezmellékletén.) Ezeket az adatokat az Excel egyik erre a célra használt külön lapján (pl. a második lapon) helyezzük el az A1:B105 tartományban. Tehát A1-be beírjuk, hogy év, B1-be azt, hogy lx, az évszámokat A2-től A105-ig, az életbenlévők számát pedig B2-től B105-ig helyezzük el.

év	lx	év	lx	év	lx
0	100000	36	94292	72	43186
1	98273	37	93950	73	40548
2	98181	38	93578	74	37849
3	98129	39	93173	75	35100
4	98096	40	92731	76	32320
5	98054	41	92247	77	29457
6	98022	42	91717	78	26671
7	97993	43	91135	79	23968
8	97964	44	90497	80	21356
9	97934	45	89801	81	18845
10	97904	46	89043	82	16446
11	97875	47	88220	83	14173
12	97847	48	87331	84	12041
13	97818	49	86372	85	10065
14	97784	50	85342	86	8259
15	97737	51	84237	87	6636
16	97682	52	83058	88	5205
17	97617	53	81805	89	3973
18	97537	54	80477	90	2940
19	97442	55	79069	91	2100
20	97333	56	77576	92	1441
21	97213	57	75991	93	945
22	97089	58	74312	94	588
23	96962	59	72537	95	345
24	96834	60	70669	96	189
25	96704	61	68714	97	96
26	96567	62	66677	98	45
27	96418	63	64565	99	19
28	96254	64	62382	100	7
29	96072	65	60134	101	2
30	95872	66	57827	102	1
31	95654	67	55476	103	0
32	95419	68	53098		
33	95167	69	50694		
34	94897	70	48253		
35	94607	71	45755		

A B oszlopban álló vektornak a biztosítási matematikában szokásos lx nevet adjuk, a következőképpen: Kijelöljük a B1:B200 tartományt, és kiadjuk az Insert Name Create parancsot TopRow beállítással. Azért mentünk el 200-ig, hogy idős ember esetén is módunk legyen akár 100 évre is előre tekinteni.

A modell felépítését az első lapon folytatjuk a fejléc és az induló paraméterek megadásával.

```
E1: S
E2: 0.4

F1: n
F2: 20

G1: x
G2: 40

H1: k
H2: 10

I1: tech.kl.
I2: 4%
```

Tehát 40 éves férfi köti a biztosítást 20 évre. Az első 10 évben 10 egyenlő részletben fizeti a biztosítási díjat. Ha 20 éven belül meghal, akkor a biztosító 1 egységnyi pénzt fizet. Ha 20 év múlva életben van, akkor a biztosító az S=0.4 összeget fizeti. A technikai kamatláb (amivel a tartalékolt összegeket kamatoztatjuk, illetve a jövőbeli pénzösszegeket jelenértékképzéskor diszkontáljuk) 4%.

A fejléc további részei:

```
B5: Évente fizetendô netto díj

B10: Még élnek
B11: P(év elején él)
B12: P(elôzô évben halt meg)
B14: P(biztosító kap)
B15: M(biztosító fizet)
```

A 9-edik sorban C9-től BU9-ig rendre elhelyezzük a 0, 1, 2, ... ,70 számokat. Ezek a számok azt jelzik, hogy a biztosítás megkötésétől számítva hányadik évben vagyunk.

C10-ben azt adjuk meg, hogy 100.000 fiú újszülöttből hány van várhatóan életben x évesen, D10-ben azt, hogy hányan érik el várhatóan az x+1 éves kort, és így tovább. A 11. sorban meghatározzuk, hogy mekkora valószínűséggel van életben a biztosított az adott év elején. A 12. sorban azt számítjuk ki, hogy mekkora valószínűséggel halt meg a biztosított éppen az előző évben.

A
```
C10: =INDEX(lx,$G$2+C9+1)
```

képlettel számítjuk ki, hogy 100.000 fiú újszülöttből x évesen hány van várhatóan életben. lx a stat. lapon lévő életbenmaradási vektor, Index(lx,i) az lx vektor i-edik koordinátája. C9 értéke 0 (később, másoláskor lesz rá szükségünk), a +1 pedig azért kell, mert az lx vektor első koordinátája a 0 évesek számát adja, s így az x évesek száma az (x+1)-edik koordináta értéke. Ezt a képletet jobbra másoljuk egészen a BU oszlopig. Eközben C9-ből D9, E9, ..., BU9 lesz, s így megkapjuk, hogy a kezdeti időpont után 1, 2, ..., 70 évvel várhatóan hányan vannak életben.

A
```
C11: =C10/$C$10
```

képletet jobbra végigmásolva megkapjuk annak valószínűségét, hogy a férfi a kezdeti időpont után pontosan 0,1,2, ..., 70 évvel még életben van.

A
```
D12: =C11 - D11
```

képlet a következőt jelenti. Akkor halt meg az illető az előző évben, ha az előző év elején még élt, de ennek az évnek az elején már nem élt. (Itt látszólag kihagytuk azt a 0 valószínűségű eseményt, amikor a biztosított másodpercnyi pontossággal éppen az előző év elején halt meg, ez azonban a számunkra szükséges várható értékeket nem befolyásolja.) Ezt a képletet is végigmásoljuk jobbra a BU oszlopig.

A 14. sorban annak valószínűségét határozzuk meg, hogy a biztosító az adott év elején megkapja a biztosítási díjat. Ennek két feltétele van: *1.* a kezdeti időponttól számított k év még nem telt le, *2.* a biztosított még életben van. A második feltétel teljesülésének valószínűsége a 11. sorban található. Ennek megfelelően;

```
C14: = =IF(C9<$H$2,C11,0)
majd másolunk jobbra
```

Itt C11 annak valószínűsége, hogy a biztosított még életben van, C9<$H$2 pedig azt mondja, hogy csak a 0., 1., ... ,k-1. év elején van fizetési kötelezettség.

A 15. sorban a biztosító által az adott év elején kifizetendő összeg várható értékét határozzuk meg.

Ez a
```
C15:=0
és
D15:=IF(D9<=$F$2,D12,0)+IF(D9=$F$2,D11
*$E$2,0)
```

képletekkel történik, az utóbbit másoljuk jobbra. A D15-be bevitt képletet így olvassuk: Ha n éven belül vagyunk, és a biztosított az előző évben halt meg, akkor a biztosító fizet 1-et, ha viszont éppen n év telt el, és a biztosított életben van, akkor a biztosító az (E2-ben lévő) S összeget fizeti.

Ezzel kiszámítottuk a biztosítási díj meghatározásához szükséges segédmennyiségeket, s rátérünk a biztosítási díj kiszámítására. A számítás a biztosítási matematikában talán legfontosabb alapelv – az ún. ekvivalencia-elv – alapján történik. Az ekvivalencia-elv szerint akkora kell hogy legyen a befizetendő díj, hogy a befizetések ekvivalensek legyenek a kifizetésekkel, azaz

A befizetések várható jelenértékeinek összege
egyenlő
a kifizetések várható jelenértékeinek összegével.

Hangsúlyozzuk, hogy itt mindkét oldalon várható értéket és jelenértéket is képzünk, tehát egy pénzösszeget mindig meg kell szorozni a bekövetkezésének valószínűségével, ezenkívül az esedékessé válás időpontjának megfelelő mértékben diszkontálni is kell. A valószínűségek a szerződéskötés pillanatában értendőek, s a jelenértékeket is erre az időpontra számítjuk. Ez az összefüggés a díjkalkuláció alapegyenlete, a szükséges nettó díjat úgy kell megállapítani, hogy ez az összefüggés teljesüljön. A szükséges valószínűségeket, illetve várható értékeket most számítottuk ki a 14. illetve 15. sorban, a diszkontálást pedig az NPV függvény fogja elvégezni az I2-ben található technikai kamatláb segítségével.

Ha a nettó díjat A-val jelöljük, akkor
A befizetések várható jelenértékeinek összege

```
A*(C14+ NPV(I2,D14:BU14))
```

továbbá

a kifizetések várható jelenértékeinek összege

```
C15+NPV(I2,D15:BU15)
```

Mivel e két összegnek azonosnak kell lennie, a nettó díj, tehát A értékét egy osztással kaphatjuk meg. Az egyszerűség kedvéért C14-et illetve C15-öt bevisszük az NPV függvényen belülre, ami azt jelenti, hogy a számlálót és a nevezőt is mégegyszer diszkontáljuk, ami természetesen nem változtatja meg a tört értékét, s ezáltal az

```
E5: =NPV(I2,C15:BU15)/NPV(I2,C14:BU14)
```

eredményhez jutunk.

40 éves férfi, n=20, k=10, 4%-os technikai kamatláb és S=0 esetén a nettó díj 0.01805-nek adódik. Ha az 1 egységnyi pénz 1 millió Forint, akkor ez azt jelenti, hogy a biztosított 10 alkalommal év elején fizet 18,050 forintot, s cserében, 20 éven belüli halál esetén egymilliót fizet a biztosító.

Tekintsük az 1 egységnyi pénzt továbbra is 1 millió Forintnak! Ha S értékét 0.4-re emeljük, tehát a 20. év elérése esetén 400,000 Ft-ot fizet a biztosító, akkor a nettó díj értéke 0.03502, tehát kb. 35 ezer forint.

A díjtartalék definíciója

Jelenlegi példánkban a biztosított 10 éven keresztül fizeti a díjat, a biztosítónak viszont 20 évig van kárenyhítési kötelezettsége. Ahhoz, hogy a biztosító a 10 év letelte után is helyt tudjon állni, az első 10 befizetésből tartalékokat kell képeznie. Ezzel eljutottunk a díjtartalék fogalmához, amit a következőképpen definiálunk:

Díjtartalék a t időpontban =
A jövőbeli kifizetések várható jelenértékeinek összege mínusz a jövőbeli befizetések várható jelenértékeinek összege.

Ehhez fontos kiegészítéseket kell tennünk. Jövő alatt ebben a cikkben a $[t,\infty)$ zárt intervallumot értjük, tehát egy t időpontbeli befizetés is beleszámít az összegzésbe. (Ha a (t,∞) nyílt intervallumot tekintenénk jövőnek, akkor egy másik díjtartalék-fogalomhoz jutnánk, amire az alább ismertetendő összefüggések változtatás nélkül nem érvényesek.) A díjtartalék időpontról időpontra változik. *A t időpontban érvényes díjtartalék kiszámításakor a várható értékeket a t időpontban érvényes valószínűségekkel* képezzük, s a *jelenértékeket is a t időpontra számítjuk*. Az egyszerűség kedvéért befizetések esetén netto díjjal számolunk, s így kizárólag az ún. netto díjtartalékkal fogunk foglalkozni. A díjtartalék a biztosító jövőbeli kötelezettségeinek fedezete, ahol a kötelezettség a kiadások és bevételek különbsége. A díjtartalék összegével a biztosító tartozik a biztosítottak közössége felé, ezért például a vállalati mérlegben a dítartalék összege a hosszúlejáratú tartozások rovatban szerepel.

Tartalmi szempontból a következőképpen lehet a díjtartalék fogalmát megvilágítani: Képzeljük el, hogy a szerződéskötés után t évvel – mondjuk átszervezés vagy profilváltás miatt – a biztosító átadja a szerződést egy másik biztosítónak. A biztosított számára semmi nem változik, ugyanazt a díjat fizeti, és ugyanabban a szolgáltatásban részesül, ugyanazokkal a feltételekkel, mint korábban. A régi biztosító az új biztosítónak nem csak az adatokat, a jogokat és a kötelezettségeket adja át, hanem egy összeget is, ami ennek a szerződésnek az *ekvivalencia-elv szerinti* teljesítéséhez szükséges. Ez az összeg a díjtartalék összege. Tehát az új biztosító megkapja a díjtartalékot, megkapja az ezután esedékes díjakat, és az ebből a két forrásból származó bevételei *ekvivalensek* lesznek a t időpont után esedékes kiadásaival. A díjtartalék végsősoron a szerződés átruházásának az ára. (Abban az esetben, ha a biztosított ruházhatná át a szerződést, tehát eladhatná egy ugyanannyi idős, ugyanolyan nemű és egészségi állapotú másik személynek, akkor a vételár lenne a díjtartalék, tehát az új biztosított ennyit fizetne a régi biztosított részére.)

A szerződés megkötése és az ezzel egyidejű első díjfizetés előtti másodpercben a díjtartalék 0 az ekvivalencia egyenlet miatt. Az első díj befizetése utáni másodpercben a díjtartalék összege az első netto díj összegével egyenlő. Ezután egy ideig növekszik, majd lehet növekvő vagy csökkenő is. A futamidő végén a díjtartalék a kifizetendő S összeggel azonos, a futamidő vége után egy másodperccel pedig 0, hiszen ekkor a jövőben már sem befizetés, sem kifizetés nem történik.

A díjtartalék kiszámítása

Ebben a pontban a vegyes biztosítás díjtartalékát közvetlenül a definíció alapján fogjuk kiszámítani. Először röviden elismételjük a vegyes biztosítási konstrukciót! x éves férfi köti a biztosítást. Ha a férfi n éven belül meghal, akkor a biztosító fizet a kedvezményezettnek egységnyi pénzt (pl. 1 millió Ft-ot). Ha viszont a biztosított n év múlva életben van, akkor ő kap a biztosítótól S összeget (pl. 400,000 Ft-ot). Ennek fejében a biztosított biztosítási díjat fizet a biztosítónak, minden év elején azonos összegben, k éven keresztül.

Az ismertetett modellel folytatjuk a munkát. Az első lapon B24-be tesszük t értékét, tehát azt, hogy hányadik év elején számítjuk a díjtartalékot. Először válasszuk t értékét 5-nek!

```
B23: Hányadik évben számoljuk a díj-
tartalékot
B24: 5
```

A 26. sorban C26-tól BU26-ig elhelyezzük a 0,1,2,3,...,70 számokat.

A fejlécet kiegészítjük az alábbi sorokkal

```
B27: t. évben még él
B28: P(még él)
B29: P(most halt meg)
```

A 26. sorban lévő számok azt jelzik, hogy a biztosítás megkötésétől számított t-edik év eleje után hányadik évben vagyunk. C27-ben azt adjuk meg, hogy 100000 fiú újszülöttből hány van várhatóan életben x+t évesen, D27-ben azt, hogy hányan érik el várhatóan az x+t+1 éves kort, és így tovább. A 28. és 29. sorban a t-edik évben (tehát x+t éves korban) érvényes valószínűségeket fogjuk meghatározni.

A

```
C27: =INDEX(lx,$B$24+$G$2+C26+1)
```

képlettel számítjuk ki, hogy 100,000 fiú újszülöttből x+t évesen hány van várhatóan életben. lx a stat lapon lévő életbenmaradási vektor, Index(lx,i) az lx vektor i-edik koordinátája. C26 értéke 0 (később, másoláskor lesz rá szükségünk), a +1 pedig azért kell, mert az lx vektor első koordinátája a 0 évesek számát adja, s így az x+t évesek száma az (x+t+1)-edik koordináta értéke. Ezt a képletet jobbra másoljuk egészen a BU oszlopig. Eközben C26-ból D26, E26, ..., BU26 lesz, s így megkapjuk, hogy a szerződés t-edik éve után 1, 2, ..., 70 évvel várhatóan hányan vannak életben.

A

```
C28: =C27/$C$27
```

képletet jobbra végigmásolva megkapjuk annak valószínűségét, hogy a férfi a szerződés t-edik éve után 0,1,2, ..., 70 évvel még életben van.

A

```
D29: =C28 - D28
```

képlettel számítjuk annak valószínűségét hogy a biztosított az x+t. évben halt meg. Ezt a képletet is végigmásoljuk jobbra a BU oszlopig, s ezzel a szerződés t-edik évében érvényes, a jövőre vonatkozó elhalálozási valószínűségekhez jutunk.

Ismét a fejléc folytatása következik.

```
B31: Bizt. ennyit kaphat
B32: Bizt. ennyit fizethet 1.
B33: Bizt. ennyit fizethet 2.
B34: M(bizt.kap)
B35: M(bizt.fizet)

B37: Díjtartalék
```

A 31. sorban azokat a díjakat határozzuk meg, amely az adott évben a biztosító számára befizetésre kerülhetnek:

```
C31: =IF($B$24+C26<$H$2,$E$5,0)
```

másolás jobbra.

Ebben a netto díj (tehát az E5 cella tartalma) szerepel, ha még nem telt le a k év.

A 32. sorban tüntetjük fel azt az összeget, melyet a biztosító ki kell hogy fizessen korai halál esetén, a 33. sorban pedig azt, amelyet életbenmaradás esetén kell kifizetnie.

```
D32: =IF($B$24+D26<=$F$2,1,0)
C33: =IF($B$24+C26=$F$2,$E$2,0)
```

Mindkettőt jobbra másoljuk. Az első szerint n éven belüli halál esetén egységnyi pénz esedékes, a második szerint az n év múlva az S összeg kifizetése esedékes. A 32. sor képlete a D oszlopban kezdődik, mert halál esetén a biztosító a halált követő évben fizet.

A 34. sorban a befizetendő összegek várható értékeit, a 35. sorban pedig a kifizetendő összegek várható értékeit számítjuk ki.

```
C34: =C28*C31
C35: =C29*C32+C28*C33
```

(másolás jobbra.)

Ezzel készen állunk a díjtartalék meghatározására. A várható értékeket már meghatároztuk, a jelenérték képzést pedig az NPV függvény végzi:

```
C37:=C35+NPV($I$2,D35:BU35)-C34-
NPV($I$2,D34:BU34)
```

Ez az érték a t időpontban érvényes díjtartalékot adja meg, ahol t a B24 cella tartalma. A képletet úgy olvassuk, hogy *a díjtartalék a t időpontban = a jövőbeli kifizetések várható jelenértékeinek összege mínusz a jövőbeli befizetések várható jelenértékeinek összege.*

40 éves férfi, n=20, k=10, 4%-os technikai kamatláb és S=0.4 esetén a nettó díj 0.03502, ugyan-akkor az 5. évben 0.20137 a díjtartalék. Ez több, mint a befizetett öt díj összege, ami nem meglepő, hiszen a korábban befizetett díjak kamatoznak.

Ahhoz, hogy a különböző t értékekhez tartozó díjtartalékokat táblázatba rendezzük, t lehetséges értékeit elhelyezzük a 39. sorban, tehát C39-től BU39-ig bevisszük a 0,1,2,...,70 számokat. Ezután a Data Table segítségével futtatjuk t értékét pl. 0-tól 30-ig, és minden t-re meghatározzuk a díjtartalékot. (30 évnél hosszab futamidejű szerződés esetén természetesen 30-nál tovább kell számolni.) Ehhez először a B40 cellába átvezetjük a díjtartalék képletét, ami csak a szerződés futamideje alatt lehet 0-tól különböző.

```
B40:
=IF(B24<$F$2,C37,0)
```

Kijelöljük a B39:AG40 tartományt, hívjuk a Data Table parancsot, s

```
Row Input Cell:
B24
```

beállítással (A Column Input Cell ablakot üresen hagyjuk, OK) megkapjuk a díjtartalékokat.

Célszerű külön lapon ábrázolni az eredményt (Data Range: C39:AG40) *(1. ábra)*

Érdemes a kockázati (S=0) biztosítás díjtartalékainak alakulását külön is megtekinteni. *(2. ábra)*

A díjtartalék rekurzív előállítása

Jelöljük most t=0,1,2,...,n esetén a t-edik év elején

a díjtartalékot DT_t-vel,

az életbenmaradás esetén a biztosító által fizetendő összeget $ÉF_t$-vel

a (t-1)-edik évben történő elhalálozás miatt a t-edik év elején a biztosító által fizetendő összeget HF_t-vel, és

a t-edik év elején befizetendő biztosítási díjat BD_t-vel.

A díjtartalék definíciójából, illetve az átruházással kapcsolatos szemléltetésből következik az alábbi rekurzív összefüggés:

$$(DT_t+BD_t ÉF_t)*(1+r)=p_{x+t}*DT_{t+1}+q_{x+t}*HF_{t+1}$$
(1)

1. ábra

Vegyes biztosítás díjtartaléka



2. ábra

Kockázati biztosítás díjtartaléka



Ebben az összefüggésben r a technikai kamatláb, px+t annak a valószínűsége, hogy a t-edik év elején x+t éves biztosított egy év múlva is életben lesz, q_{x+t} pedig annak a valószínűsége, hogy ugyanez a biztosított a t-edik évben meg fog halni. Az összefüggés azt mondja ki, hogy a t időpontban rendelkezésre álló DT_t díjtartalékhoz hozzáadódik a BD_t befizetett díj, ugyanakkor az így kapott összegből le kell vonni az életbenmaradás esetén kifizetendő ÉFt összeget. A megmaradt rész egy éven keresztül kamatozik, s így egy év múlva életbenmaradás esetén rendelkezésre áll az új díjtartalék, halál esetén pedig rendelkezésre áll a halál esetén fizetendő összeg. Az összefüggés segítségével a $DT_0=0$ egyenlőségből kiindulva rekurzív módon sorra kiszámíthatjuk a díjtartalékokat a t=1,2,...,n értékekre.

(2)

$DT_{t+1}=((DT_t+BD_t-ÉFt)*(1+r)-q_{x+t}*HF_{t+1})/p_{x+t}$

Az Excel modell folytatásaként ellenőrizzük először az (1) összefüggést. A fejlécet az alábbi módon bővítjük:

```
B42: Befizetett díj
B43: Élet esetén kap
B44: Halál esetén kap
B45: P(mégegy évet él)
B46: P(most fog meghalni)
```

A nettó díj értéke az E5 cella tartalma, és ezt k alkalommal kell befizetni, ezért

```
C42: =IF(C39<$H$2,$E$5,0)
```

Ha a biztosított n év múlva életben van, akkor az E2-ben lévő S összeg kerül kifizetésre, tehát

```
C43: =IF(C39=$F$2,$E$2,0)
```

A C42-es és C43-as képleteket jobbra másoljuk a BU oszlopig.

Halál esetén a rákövetkező év elején esedékes a kifizetés, így

```
C44: 0          D44: =IF(C39=$F$2,$E$2,0)
```

és a D44-es képletet másoljuk jobbra.

Annak a valószínűségét, hogy a biztosított az előtte álló év végén még életben lesz, a

```
C45: =D10/C10
```

képlettel számítjuk, míg annak a valószínűsége, hogy a biztosított épp az előtte álló évben fog meghalni,

```
C46: =1-C45
```

Mindkét képletet jobbra másoljuk.

Ezekután ki tudjuk számítani az (1) összefüggés két oldalán álló mennyiségeket.

```
B48: baloldal
B49: jobboldal

C48: =(C40+C42-C43)*(1+$I$2)
C49: =C45*D40+C46*D44
```

Miután mindkét képletet jobbra másoljuk, természetesen azt tapasztaljuk, hogy a 48. és 49. sorban azonos számok állnak.

A (2) képlet segítségével, most egy másik, rekurzív módszer segítségével is kiszámítjuk a dítartalékokat.

```
B50: Díjtartalék rekurzióval
C50: 0

D50:=((C50+C42-C43)*(1+$I$2)-
C46*D44)/C45
```

A D50 képletet jobbra végigmásolva természetesen azt látjuk, hogy az 50. sorban keletkező számok a 40. sor számaival azonosak. A két kiszámítási módszer közül a rekurzív eljárás egyszerűbb, kevesebb számolást igényel, s egy jobb képességű (némi memóriával rendelkező) zsebszámológéppel, vagy akár kockás papíron is elvégezhető. A 40. sorban alkalmazott eljárás a korszerű táblázatkezelő rendszerek magasszintű szolgáltatásaira (Data Table technika, pénzügyi függvények stb.) épít. Ugyanakkor közvetlenül a definíciót használja fel, ami komoly előny. Érdemes még azt is hozzátenni, hogy egy korszerű táblázatkezelő program használata ma már a pénzügyi életben alapkövetelmény.

A biztosítási díj komponensei

Az (1) összefüggésből nemcsak az új díjtartalék fejezhető ki, hanem maga a biztosítási díj is. Felhasználva a

$p_{x+t}= 1 - q_{x+t}$

összefüggést

$BD_t=DT_{t+1}/(1+r)-DT_t+ÉF_t+q_{x+t}*(HF_{t+1}-DT_{t+1})/(1+r)$

adódik. Ennek az összefüggésnek a jobboldalán a biztosítási díj két komponense látható. Az első komponens a díjtartalék változásának és az életbenmaradás esetén történő helytállásnak a fedezete, a második komponens pedig a halál esetén a díjtartalékot meghaladó összeggel történő helytálláshoz szükséges.

1. komponens = $DT_{t+1}/(1+r) - DT_t + ÉF_t$

2. komponens = $q_{x+t} *(HF_{t+1} - DT_{t+1})/(1+r)$

A t-edik év elején ki kell fizetni az $ÉF_t$ összeget, továbbá gondoskodni kell a díjtartalék változásának fedezetéről. Erre szolgál az 1. komponens. Halálozás esetén a fizetendő összeg általában nagyobb, mint a DT_{t+1} díjtartalék. A különbség fedezete a második komponens. Az első komponenst felhalmozási résznek, a másodikat kockázati résznek nevezzük. A kettő összege a befizetett díj.

Excel modellünkben a következőképpen határozzuk meg ezeket a komponenseket.

```
B51: Kockázati rész
B52: Felhalmozási
     rész

C51: =C46*(D44-
D40)/(1+$I$2)
C52: =D40/(1+$I$2)-
C40+C43
```

A C51 és C52 képleteket jobbra másoljuk. A két komponens összege az első 10 évben természetesen az E5-ben található biztosítási díjjal azonos, utána pedig 0. Célszerű a két komponenst grafikonon ábrázolni. *(3. ábra)*

Az ábrán jól látható, hogy a kockázati rész fokozatosan emelkedik, hiszen az életkor előrehaladásával a halálozás valószínűsége növekszik. Az első 10 évben a biztosítási díj konstans, ennek megfelelően a másik komponens, tehát a felhalmozási rész, csökken. Amikor a díjfizetés véget ér, a kockázati rész tovább növekszik. Ugyanakkor a „felhalmozási rész" negatív lesz, hiszen a két komponens összege 0. Ilyenkor inkább felélési részről érdemes beszélni, ebben az időszakban ugyanis a díjfizetés 0, s a növekvő kockázati rész kompenzációjaként a biztosított a korábban felhalmozott díjtartalékot fokozatosan feléli. Érdemes még megjegyezni, hogy tisztán kockázati biztosítás (tehát S=0) esetén a két komponens közül a kockázati rész lényegesen nagyobb szerepet játszik, amint azt a *4. ábra* is mutatja. *(4. ábra)*

3. ábra

A biztosítási díj komponensei



4. ábra

A biztosítási díj komponensei

s=0 esetén



Irodalomjegyzék

Banyár József: Az életbiztosítás alapjai. Bankárképző – Biztosítási Oktatási Intézet, Budapest, 1994

Fiala Tibor: Pénzügyi modellezés EXCEL-lel. Kossuth Kiadó, Budapest, 1999

Getting Results with Microsoft Office 97. Microsoft Corporation, 1995–96

Kovalcsik Géza: EXCEL'97. ComputerBooks, 1997

Krekó Béla: Életbiztosítás I. Budapesti Közgazdaságtudományi Egyetem, AULA, 1994

K. Wolfsdorf: Versicherungsmathematik. B. G. Teubner, Stuttgart, 1997

Vetier András: Szemléletes mérték- és valószínűségelmélet. Nemzeti Tankönyvkiadó, Budapest, 1991